# $Id: PKGBUILD 86268 2013-03-14 14:36:13Z bpiotrowski $
# Maintainer:  Bartłomiej Piotrowski <barthalion@gmail.com>
# Contributor: kfgz <kfgz at interia dot pl>
# Contributor: pootzko <pootzko at gmail dot com>

pkgname=haveged
pkgver=1.7.a
_pkgver=${pkgver/.a/a}
pkgrel=1
pkgdesc='Entropy harvesting daemon using CPU timings'
arch=('i686' 'x86_64')
url='http://www.issihosts.com/haveged'
license=('GPL')
depends=('glibc')
source=($url/$pkgname-$_pkgver.tar.gz
        service)
sha256sums=('8e7b0779bd8a4b9b73769ddaf626918355e2a542d3ac35e5ec31b74a58b9b051'
            '85aed005b7d0d0bae74f7fc0ca70907a6f67ddbcb2df77ca004f9a919028564a')

build() {
  cd $srcdir/$pkgname-$_pkgver
  ./configure --prefix=/usr
  make
}

package() {
  cd $srcdir/$pkgname-$_pkgver
  make DESTDIR=$pkgdir install

  rm -rf $pkgdir/etc/init.d
  install -D -m644 $srcdir/service $pkgdir/usr/lib/systemd/system/haveged.service
}
